

Quatrième partie

Ecrire une application
pour le
NANORESEAU

Comment développer une application pour le NANORESEAU

Ce chapitre s'adresse tout particulièrement aux développeurs désireux d'écrire des applications pour le NANORESEAU. Tous les exemples seront donnés en Assembleur 6809. C'est d'ailleurs en Assembleur qu'ont été écrits les principaux langages ou applications pour le MO5 ou le TO7 en réseau (BASIC, LOGO, LSE, COLORPEINT, CARACTOR, FORMATOR, etc.).

Il n'existe pas encore, à l'heure actuelle, d'Assembleur 6809 fonctionnant sur le MO5 ou le TO7 en réseau et permettant d'écrire directement des applications complètes. Les développements ne peuvent donc se faire que sur une machine plus importante et être transférés sur les machines cibles (MO5 ou TO7).

On peut, bien entendu, utiliser une très grosse machine, comme un VAX, sur laquelle il existe de puissants outils de développement. Si ce genre de système a été effectivement utilisé pour écrire un logiciel pour le réseau, ce n'est malheureusement pas à la portée de tout le monde...

Une autre solution consiste à utiliser un micro-ordinateur fonctionnant avec un 6809 et possédant un bon éditeur et un bon Assembleur. C'est, en particulier, le cas du Goupil 3 PC 6809. Tout le travail d'édition et d'assemblage peut se faire sur cette machine. Le résultat obtenu doit ensuite être transféré sur une disquette au format MS-DOS pour pouvoir être lu par le central du NANORESEAU et donc par les postes.

Enfin on peut aussi envisager l'utilisation d'un Assembleur fonctionnant sous MS-DOS et produisant du code 6809. On trouve de la publicité pour de tels "cross-assembleurs" dans la presse spécialisée (le mensuel BYTE par exemple) à un prix tout à fait raisonnable. Dans cette solution, le central du NANORESEAU sert alternativement de machine de développement et de gestionnaire de réseau.

Quels que soient les outils de développement utilisés, l'application peut être livrée dans une ROM (une MEMO5 ou une MEMO7) ou dans un fichier MS-DOS pour être téléchargée dans les postes à partir du central.

Dans le contexte NANORESEAU, les fichiers contenant des applications doivent avoir une structure bien particulière :

- Ils doivent, comme tout fichier réseau, avoir un en-tête proprement constitué. La structure de cet en-tête est donnée en annexe et il est très possible de la faire générer par l'Assembleur (voir l'exemple du paragraphe suivant).

- Ils doivent, de plus, avoir un format binaire simple ou un format binaire étendu. La description de ces deux formats est également donnée en annexe. Le

format binaire étendu est plus général que le format binaire simple. Il permet de gérer les différentes banques de mémoire des postes (cartouche de RAM en particulier) et autorise le téléchargement de l'application depuis le central. C'est pourquoi il est recommandé de l'utiliser. Les exemples donnés dans la suite ont le format binaire étendu.

- Enfin, les fichiers contenant des applications doivent avoir l'extension .MO5 s'ils sont destinés au MO5 et .TO7 s'ils sont destinés au TO7.

MINIDOS et MINIRES

Pour faciliter le travail des développeurs d'applications destinées au MO5 et au TO7 et fonctionnant avec un lecteur de disquettes, les ingénieurs de Thomson ont écrit un ensemble de sous-programmes qui constituent un système d'exploitation des fichiers sur disquettes compatible avec la gestion des fichiers de BASIC. Cet ensemble est connu sous le nom de MINIDOS. Il est fourni par Thomson aux développeurs qui lui en font la demande. C'est ainsi que la quasi-totalité des cartouches qui fonctionnent avec un lecteur de disquettes contiennent MINIDOS.

MINIDOS réalise une gestion séquentielle des fichiers, la suppression d'un fichier et l'affichage du catalogue des disquettes. Ces fonctionnalités sont largement suffisantes dans la majorité des cas.

Pour faciliter le passage des applications sur le NANORESEAU, une adaptation de MINIDOS aux primitives du système d'exploitation du NANORESEAU a été faite et porte le nom de MINIRES. MINIRES offre les mêmes points d'entrée que MINIDOS avec les mêmes modalités d'appel. Pour qu'une application fonctionnant avec le lecteur de disquettes fonctionne directement sur le réseau, il suffit souvent de remplacer tout simplement MINIDOS par MINIRES. MINIRES est fourni par Thomson aux développeurs qui lui en font la demande.

Il faut toutefois souligner que MINIRES ne propose qu'une gestion séquentielle des fichiers. Pour ceux qui voudraient utiliser toute la richesse et la puissance des primitives du système d'exploitation du NANORESEAU, il n'y a qu'une seule solution : employer ces primitives directement.

Un exemple d'application dialoguant avec le central

On trouvera à la fin de ce chapitre le listing d'une petite application nommée LISTE.ASM qui se contente d'afficher à l'écran le contenu d'un fichier puis de retourner au niveau système. Cette petite application a un triple intérêt :

- Elle permet de mettre en évidence les initialisations que l'on doit faire dans toute application.
- Elle donne un exemple de constitution par l'Assembleur de l'en-tête réseau et du format binaire étendu.
- Elle montre enfin la façon d'appeler le central pour utiliser les primitives du système d'exploitation du NANORESEAU.

Cette application peut fonctionner dans le MO5 ou dans le TO7 en réseau. Un paramètre d'assemblage permet de faire une version destinée à l'une ou à l'autre de ces machines.

A l'entrée dans l'application, il faut faire en sorte que la page 0 du moniteur soit "propre". Comme notre application peut être chargée depuis le central, elle peut arriver dans un poste après une application qui aurait modifié certains pointeurs de cette page 0. Si ces pointeurs ne sont pas remis à leurs valeurs standard, cela peut provoquer des catastrophes. La meilleure façon de nettoyer la page 0 du moniteur est de provoquer un faux RESET; c'est alors le moniteur lui-même qui fait le travail ! Pour pouvoir reprendre la main, il suffit d'initialiser correctement le point d'entrée à froid dans l'application nommé COLD dans notre exemple. Tout cela ne peut fonctionner que si le poste est pourvu d'une cartouche de RAM.

Il faut prévoir le fait que l'utilisateur peut appuyer sur le bouton RESET qui est très facile d'accès sur le MO5 ou sur le TO7. En d'autres termes, il faut pouvoir reprendre la main après un vrai RESET en mettant dans le point d'entrée à chaud une adresse appropriée. Si le poste dispose d'une cartouche de RAM, il faut initialiser le point d'entrée à chaud qui se trouve dans chaque banque de la cartouche de RAM. En effet, on ne peut pas savoir à l'avance quelle banque de la cartouche sera visible au moment où l'utilisateur déclenchera le RESET.

De toutes les façons, après un vrai ou un faux RESET, il est impératif d'exécuter les six instructions suivantes qui réinitialisent le poste dans le réseau :

		MO5	TO7
LDX	#IRQRE	\$A041	\$E041
STX	IRQPT	\$2064	\$6064
LDA	NUPOST	\$A7D8	\$E7D8
ANDA	#\$1F		
STA	NUPO	\$2052	\$6052
JSR	RELACH	\$A02B	\$E02B

Après avoir nettoyé la page 0 du moniteur et réinitialisé le poste dans le réseau, il faut mettre à jour trois pointeurs de la RAM réseau (ACTRLC, ADCRDU et ATRCTR).

Et l'application peut alors vraiment démarrer...

Les quatre primitives du système d'exploitation du NANORESEAU utilisée dans cet exemple sont :

- Ouvrir un fichier en lecture
- Lire dans un fichier
- Fermer un fichier
- Revenir au niveau du système d'exploitation

Insistons sur le fait que cette dernière primitive doit être la voie de sortie normale de toute application fonctionnant dans le NANORESEAU. Sa mise en œuvre est très simple. Elle a pour effet de terminer proprement l'application et d'amener dans le poste le module système des postes (NR-DOS par défaut).

; LISTE.ASM - GILBERT CORNILLIET - 08 12 85

; CETTE MINI-APPLICATION DESTINEE AU M05 OU AU T07 EN NANOSEAU AFFICHE
; LE CONTENU DU FICHIER TOTO.ABC SUR L'ECRAN DU POSTE ET SE TERMINE
; PAR LA DEMANDE DE CHARGEMENT DU MODULE SYSTEME.

; CET EXEMPLE ILLUSTRE

; * LA FACON DONT DOIT ETRE CONSTITUEE UNE APPLICATION NANOSEAU ET EN
; PARTICULIER LA MANIERE DE FAIRE FABRIQUER PAR L'ASSEMBLEUR UN FICHIER
; AU FORMAT BINAIRE ETENDU.

; * LES INITIALISATIONS QU'IL CONVIENT DE FAIRE AU LANCEMENT DE
; L'APPLICATION.

; * LA MANIERE D'APPELER LE RESEAU ET PLUS PARTICULIEREMENT LE GESTIONNAIRE
; DE FICHIERS AU CENTRAL.

; L'APPLICATION EN ELLE-MEME (AFFICHAGE DU CONTENU D'UN FICHIER A L'ECRAN)
; EST RUDIMENTAIRE CAR CE N'EST PAS L'ESSENTIEL ICI.

FF FF	VRAI	EQU	-1	
00 00	FAUX	EQU	0	
00 00	T07	EQU	FAUX	
FF FF	M05	EQU	VRAI	
00 95	XEOF	EQU	149	; ERREUR "FIN DE FICHIER"
00 00	CENTRAL	EQU	0	; LE CENTRAL EST CONSIDERE COMME LE POSTE 0
00 09	SYSFIC	EQU	9	; CODE TACHE RESEAU DU GESTIONNAIRE DE FICHIERS
				; LES PRIMITIVES DU GESTIONNAIRE DE FICHIERS
00 24	OUVFL	EQU	36	; OUVERTURE DE FICHIER EN LECTURE
00 28	LIRFI	EQU	40	; LECTURE DANS UN FICHIER
00 2A	FERFI	EQU	42	; FERMETURE DE FICHIER
00 32	SYSTEM	EQU	50	; RETOUR AU NIVEAU DU SYSTEME

	IF	M05		
50 00	ORIGINE	EQU	\$5000	; L'ADRESSE D'IMPLANTATION DU CODE
50 00	GOTO	EQU	\$5000	; L'ADRESSE D'EXECUTION
00 00	XPAGE	EQU	\$00	; PAS DE COMMUTATION DE BANQUE
	ENDIF			; M05
	IF	T07		
	ORIGINE	EQU	\$9000	; L'ADRESSE D'IMPLANTATION DU CODE
	GOTO	EQU	\$9000	; L'ADRESSE D'EXECUTION
	XPAGE	EQU	0	; PAS DE COMMUTATION DE BANQUE
	ENDIF			; T07

;---- SECTEUR D'EN-TETE

```

4F54          ORG   ORIGINE-172   ; LA PLACE DE L'EN-TETE ET DU RESTE

4F54 2A 4E 52 55   DEBBIN: DB   'NRUSTL*'
4F58 53 54 4C 2A
4F5C 20 20 20 20   DB   '      '   ; IDENTIFICATION DU FICHIER (STANDARD)
4F60 20 20 20 20

4F64 01          DB   1           ; UN OCTET AYANT UNE VALEUR FIXE : 1
4F65 02          DB   2           ; TYPE DU FICHIER : LANGAGE MACHINE
4F66 00          DB   0           ; MODE : BINAIRE

; LONGUEUR DU FICHIER SUR 4 OCTETS ( CONVENTION Z80 - 8088 )

4F67 C6          DB   (FINBIN-DEBBIN).MOD.256
4F68 01          DB   (FINBIN-DEBBIN)/256
4F69 00          DB   0

4F6A 00          DB   0           ; UN OCTET NUL

4F6B 00          DB   0           ; FICHIER EN LECTURE ET EN ECRITURE
4F6C 03 03       DB   3,3        ; VERSION 3.3 DU NANOSEAU

4F6E 55 0C 08    DB   85,12,8     ; DATE DE CREATION
4F71 55 0C 08    DB   85,12,8     ; DATE DE MISE A JOUR

IF M05
4F74 01          DB   1           ; ORDINATEUR AYANT CREE LE FICHIER
ENDIF ; M05

IF T07
DB 0
ENDIF

4F75 00          DB   0           ; LANGAGE DE L'APPLICATION AYANT CREE CE FICHIER
; 0 CORRESPOND A UNE APPLICATION NON REFERENCEE

4F76            DS   46          ; 46 OCTETS RESERVES AU SYSTEME POUR EXTENSIONS

4FA4            DS   48          ; 48 OCTETS DESTINES AUX APPLICATIONS

```

;---- LA SIGNATURE DU FICHIER

```

4FD4 01          DB   1           ; TYPE DE L'ENREGISTREMENT : SIGNATURE
;
; TYPE DE MACHINE
; 0 = T07
; 1 = M05
; 2 = T07 70

IF M05
4FD5 01          DB   1           ; M05
ENDIF ; M05

IF T07
DB 0
ENDIF ; T07

```

```

4FD6 00          DB      0          ; CODE DU LANGAGE
;              0 = NON REFERENCE
;              1 = BASIC
;              2 = LOGO
;              3 = LSE

4FD7 C0          DB      $C0        ; OCTET DE CHARGEMENT
;              BIT 7 A 1 : ON PEUT TELECHARGER CE
;                          FICHER DEPUIS LE CENTRAL
;              BIT 6 A 1 : PRISE EN COMPTE DE L'OCTET
;                          LANGAGE

4FDB 00 00       DB      0,0        ; DEUX OCTETS NULS

4FDA 4C 49 53 54 DB      'LISTE',0,0,0 ; LE NOM DE CETTE APPLICATION
4FDE 45 00 00 00
4FE2 00 00 00 00 DB      0,0,0,0,0,0,0,0
4FE6 00 00 00 00
4FEA 00 00 00 00 DB      0,0,0,0,0,0,0,0
4FEE 00 00 00 00
4FF2 00 00 00 00 DB      0,0,0,0,0,0,0,0
4FF6 00 00 00 00

```

;---- LE CODE

```

4FFA 02          DB      2          ; C'EST DU CODE !
4FFB 01 14       DW      FINPRO-DEBPRO ; SA LONGUEUR
4FFD 50 00       DW      DEBPRO      ; L'ADRESSE D'IMPLANTATION
4FFF 00          DB      XPAGE      ; LA PAGE
5000

```

;---- LE PROGRAMME

```

5000          DEBPRO:

;
;
IF      T07

PILE      EQU      $6200          ; LA PILE
COLD      EQU      $001E          ; POINT D'ENTREE A FROID DANS LA ROM
WARM      EQU      $001C          ; POINT D'ENTREE A CHAUD DANS LA ROM
PORT      EQU      $E7C3          ; POUR CHANGER DE PAGES ECRAN
RESEAU    EQU      $E025          ; APPEL AU RESEAU
CTRLC     EQU      $5FFE          ; REPRISE EN MAIN APRES <CTRL-C>
ATRCTR    EQU      $5FFA          ; TRAITEMENT CODE TACHE APPLICATION
ADCRDU    EQU      $5FF4          ; ADRESSE DU BUFFER DE COMPTE RENDU
NUPO      EQU      $6052          ; NUMERO DE POSTE DANS PAGE ZERO MONITEUR
NUPOST    EQU      $E7DB          ; NUMERO DE POSTE PAR LECTURE DES INTERUPTEURS
IRQRE     EQU      $E041          ; ADRESSE DE LA ROUTINE DE RECEPTION SOUS IRQ
IRQPT     EQU      $6064          ; POINTEUR DES IRQ UTILISATEUR
RELACH    EQU      $E02B          ; RELACHE DE LIGNE .

ENDIF      ; T07

```

```

                IF      M05

FF FE          RESET EQU  $FFFEH      ; POINT D'ENTREE DU RESET

22 00          PILE  EQU  $2200        ; LA PILE
EF FE          COLD  EQU  $EFFF        ; POINT D'ENTREE A FROID DANS LA ROM
EF FE          WARM  EQU  $EFFF        ; POINT D'ENTREE A CHAUD DANS LA ROM
A7 C0          PORT  EQU  $A7C0        ; POUR CHANGER DE PAGES ECRAN
A0 25          RESEAU EQU  $A025        ; APPEL AU RESEAU
1F FE          ACTRLC EQU  $1FFE        ; REPRIS EN MAIN APRES <CTRL-C>
1F FA          ATRCTR EQU  $1FFA        ; TRAITEMENT CODE TACHE APPLICATION
1F F4          ADCRDU EQU  $1FF4        ; ADRESSE DU BUFFER DE COMPTE RENDU
20 52          NUPO  EQU  $2052        ; NUMERO DE POSTE DANS PAGE ZERO MONITEUR
A7 DB          NUPOST EQU  $A7DB        ; NUMERO DE POSTE PAR LECTURE DES INTERRUPTEURS
A0 41          IRQRE EQU  $A041        ; ADRESSE DE LA ROUTINE DE RECEPTION SOUS IRQ
20 64          IRQPT EQU  $2064        ; POINTEUR DES IRQ UTILISATEUR
A0 2B          RELACH EQU  $A02B        ; RELACHE DE LIGNE

                ENDIF ; M05

                ; LE CODE DE L'APPLICATION

5000           DEBUT:

                IF      M05

                ; LES SIX INSTRUCTIONS SUIVANTES SERVENT A NETTOYER LA
                ; PAGE ZERO DU MONITEUR. EN EFFET, CETTE APPLICATION POUVANT
                ; ETRE TELECHARGEE DEPUIS LE CENTRAL, IL POURRAIT SE FAIRE QUE
                ; CERTAINS POINTEURS DE LA PAGE ZERO DU MONITEUR NE SOIENT PAS
                ; A JOUR.

5000 1A FF          ORCC  ##FF          ; MASQUER LES INTERRUPTIONS

5002 8E 50 11       LDX   #RETOUR
5005 BF EF FE       STX   COLD

5008 BC EF FE       CMPX  COLD          ; ET S'IL N'Y AVAIT PAS DE CARTOUCHE DE RAM ?
500B 26 04         BNE   RETOUR

500D 6E 7F FF FE   JMP   [RESET]      ; SIMULATION D'UN RESET. LE MONITEUR VA
                                        ; REINITIALISER LE POSTE CORRECTEMENT.

5011           RETOUR:

                ENDIF ; M05

                ; ON VIENT DE FAIRE UN TOUR DANS LE MONITEUR...

5011 8E 50 C0       LDX   #COLOR        ; VIDER L'ECRAN ET DONNER LES COULEURS
5014 BD 50 AA       JSR   MES

5017 1A FF          ORCC  ##FF          ; MASQUER LES INTERRUPTIONS

5019 10 CE 22 00   LDS   #PILE

```

; APRES TOUT RESET VRAI OU SIMULE (VOIR CI-DESSUS), IL FAUT
 ; REINITIALISER LE POSTE DANS LE RESEAU EN EXECUTANT LA SUITE
 ; D'INSTRUCTIONS SUIVANTE :

501D	8E A0 41	LDX	#IRQRE	; METTRE LE POINTEUR D'IRQ SUR LA ROUTINE
5020	BF 20 64	STX	IRQPT	; DE RECEPTION DE L'EPROM RESEAU. ; EN EFFET, LE MONITEUR A MODIFIE CE POINTEUR
5023	B6 A7 D8	LDA	NUPOST	; REMETTRE LE NUMERO DU POSTE DANS LA PAGE ZERO
5026	B4 1F	ANDA	#%1F	
502B	B7 20 52	STA	NUPO	
502B	BD A0 2B	JSR	RELACH	; RELACHE DE LIGNE
				; ET MAINTENANT, IL FAUT INITIALISER QUELQUES POINTEURS NECESSAIRES ; AU FONCTIONNEMENT DU POSTE DANS LE RESEAU.
502E	BD 50 B3	JSR	PAGECA	; SE METTRE EN PAGE CARACTERE
5031	8E 50 00	LDX	#DEBUT	; REPRISE EN MAIN APRES <CTRL-C>
5034	BF 1F FE	STX	ACTRLC	; OU NOMBRE D'ESSAIS EPUISE
5037	8E 51 0B	LDX	#CRDU0	; ADRESSE DU BUFFER DE COMPTE RENDU
503A	BF 1F F4	STX	ADCRDU	
503D	8E 50 BF	LDX	#TRCTR	; TRAITEMENT DES CODES TACHES RESEAU
5040	BF 1F FA	STX	ATRCTR	; INCONNUS DE L'EPROM RESEAU
				; ET ON AUTORISE A NOUVEAU LES INTERRUPTIONS
5043	1C 00	ANDCC	#%00	
				; C'EST ICI QUE COMMENCE LE CODE SPECIFIQUE A NOTRE MINI-APPLICATION ; OUVERTURE DU FICHIER EN LECTURE
5045	8E 50 CE	LDX	#ZCOUV	
504B	BD A0 25	JSR	RESEAU	
504B	F6 51 0C	LDB	CRDU	; ERREUR A L'OUVERTURE ?
504E	26 42	BNE	ERROR	
				; CA S'EST BIEN PASSE. ON RECUPERE LE NUMERO LOGIQUE DU FICHIER
5050	F6 51 0D	LDB	CRDU+1	
5053	F7 50 F0	STB	NUL061	
5056	F7 50 FF	STB	NUL062	
				; LECTURE DU FICHIER DANS NOTRE BUFFER BUFF
5059	8E 50 E5	LDX	#ZCLEC	
505C	BD A0 25	JSR	RESEAU	
505F	F6 51 0C	LDB	CRDU	
5062	27 06	BEQ	LIST10	
5064	C1 95	CMPB	#XEOF	; IL FAUT FILTRER LA FAUSSE ERREUR
5066	27 02	BEQ	LIST10	; "FIN DE FICHIER".
506B	20 2B	BRA	ERROR	

```

; ON AFFICHE CE QU'ON VIENT DE LIRE

506A 10 BE 51 0D LIST10: LDY CRDU+1 ; Y CONTIENT LE NOMBRE DE CARACTERES A AFFICHER
; ON SUPPOSE QUE LA LONGUEUR DU FICHIER
; EST INFERIEURE A 64 KILO-OCTETS

506E 27 0F BEQ LOOPEND ; SI LE FICHIER A UNE LONGUEUR NULLE ...

5070 8E 51 14 LDX #BUFF ; X POINTERA DANS LE BUFFER

5073 E6 B0 LOOP: LDB ,X+ ; CARACTERE SUIVANT

5075 C1 1A CMPB #1A ; FIN DE FICHIER STYLE CPM ET MS-DOS ?
5077 27 06 BEQ LOOPEND

5079 8D 2D BSR CD ; NON. ON AFFICHE DONC LE CARACTERE
507B 31 3F LEAY -1,Y
507D 26 F4 BNE LOOP ; ET ON CONTINUE S'IL RESTE DES CARACTERES

507F LOOPEND:

; FERMETURE DU FICHIER

507F BE 50 F4 LDX #ZCFER
5082 BD A0 25 JSR RESEAU
5085 F6 51 0C LDB CRDU ; ERREUR A LA FERMETURE ?
5088 26 08 BNE ERROR

; RETOUR AU SYSTEME

508A BE 51 00 FIN: LDX #ZCSYS
508D BD A0 25 JSR RESEAU

; ON NE DOIT PAS EN REVENIR ...

5090 20 FE BRA $

; AFFICHAGE D'UN MINI-MESSAGE D'ERREUR

5092 BE 50 9A ERROR: LDX #MESERR
5095 BD 50 AA JSR MES
5098 20 F0 BRA FIN

509A 0D 0A 45 72 MESERR: DB 13,10,'Erreur ...',7,0
509E 72 65 75 72
50A2 20 2E 2E 2E
50A6 07 00

```

```

; <--- CD - AFFICHAGE D'UN CARACTERE SUR L'ECRAN
; E) B CONTIENT LE CARACTERE A AFFICHER
; S -
; LE TRAITEMENT EST FAIT PAR LE MONITEUR DU M05 OU DU T07
; >---

50AB CO:
50AB 3F IF M05
50A9 B2 SWI
DB B2H
ENDIF ; M05

IF T07
JMP OE803H
ENDIF ; T07

; <--- MES - AFFICHAGE D'UN MESSAGE A L'ECRAN
; E) X POINTE SUR UNE ZONE MEMOIRE TERMINEE PAR UN CARACTERE NUL
; S) -
; >---

50AA E6 80 MES: LDB ,X+
50AC 26 01 BNE MES0
50AE 39 RTS

50AF 8D F7 MES0: BSR CO
50B1 20 F7 BRA MES

; <--- PAGECA - PASSAGE EN PAGE CARACTERE
; E) -
; S) SEUL CC MODIFIE
; ON PEUT ACCEDER A LA PAGE CARACTERE DE L'ECRAN
; >---

50B3 34 04 PAGECA: PSHS B
50B5 F6 A7 C0 LDB PORT
50B8 CA 01 ORB #1
50BA F7 A7 C0 STB PORT
50BD 35 B4 PULS B,PC

; <--- TRCTR - TRAITEMENT DES CODES TACHES RESEAU INCONNUS DE L'EPROM
; E) CA VIENT DU LOGICIEL DE COMMUNICATION
; S) ON NE FAIT RIEN. LE LOGICIEL DE COMMUNICATION SE CHARGERA DE TERMINER
; LE DIALOGUE.
; >---

50BF 39 TRCTR: RTS

```

;----- LES DONNEES -----

; APRES UN RESET, ON EFFACE L'ECRAN ET ON INITIALISE LES COULEURS

50C0				
50C0	1F 20 20	COLOR:	DB \$1F,\$20,\$20	; HAUT DE LA FENETRE : LIGNE 0
50C3	1F 12 14		DB \$1F,\$12,\$14	; BAS DE LA FENETRE : LIGNE 24
50C6	1B 44		DB \$1B,\$44	; FORME BLEUE
50C8	1B 56		DB \$1B,\$56	; FOND CYAN
50CA	1B 66		DB \$1B,\$66	; TOUR CYAN
50CC	0C		DB \$0C	; EFFACEMENT DE L'ECRAN
50CD	00		DB 0	

; LA ZONE CONSIGNE POUR L'OUVERTURE DU FICHER

50CE	18	ZCOUV:	DB 24	; LONGUEUR DE CETTE ZONE CONSIGNE
50CF	00		DB CENTRAL	; APPEL AU CENTRAL
50D0	09		DB SYSFIC	; LE GESTIONNAIRE DE FICHERS
50D1	24		DB OUVFL	; OUVERTURE DE FICHER EN LECTURE
50D2	00 00		DB 0,0	; NOMBRE D'OCTETS
50D4	00		DB 0	; PAGE
50D5	00 00		DB 0,0	; ADRESSE
50D7	00		DB 0	; TYPE D'ORDINATEUR
50D8	00		DB 0	; CODE DE L'APPLICATION
50D9	01 54 4F 54		DB 1,'TOTO ABC'	; LE NOM DU FICHER A OUVRIR
50DD	4F 20 20 20			
50E1	20 41 42 43			

; LA ZONE CONSIGNE POUR LA LECTURE DANS LE FICHER

50E5	10	ZCLEC:	DB 16	; LONGUEUR DE LA ZONE CONSIGNE
50E6	00		DB CENTRAL	; APPEL AU CENTRAL
50E7	09		DB SYSFIC	; LE GESTIONNAIRE DE FICHERS
50E8	28		DB LIRF1	; LECTURE DANS UN FICHER
50E9	9E EC		DW \$F000-BUFF	; LA LONGUEUR DU BUFFER DE RECEPTION
50EB	00		DB 0	; PAGE
50EC	51 14		DW BUFF	; L'ADRESSE POUR RECEVOIR LES DONNEES
50EE	00		DB 0	; TYPE D'ORDINATEUR
50EF	00		DB 0	; CODE DE L'APPLICATION
50F0	00	NULOG1:	DB 0	; LE NUMERO LOGIQUE DU FICHER OUVERT
50F1	00 00 00		DB 0,0,0	; L'ADRESSE DANS LE FICHER

```

; LA ZONE CONSIGNE POUR LA FERMETURE DU FICHIER

50F4 0C          ZCFER: DB 12          ; LONGUEUR DE CETTE CONSIGNE

50F5 00          DB  CENTRAL        ; APPEL AU CENTRAL
50F6 09          DB  SYSFIC         ; LE GESTIONNAIRE DE FICHIERS
50F7 2A          DB  FERFI          ; FERMETURE DE FICHIER

50F8 00 00       DB  0,0           ; NOMBRE D'OCTETS
50FA 00          DB  0             ; PAGE
50FB 00 00       DB  0,0           ; ADRESSE

50FD 00          DB  0             ; TYPE D'ORDINATEUR
50FE 00          DB  0             ; CODE DE L'APPLICATION

50FF 00          NULOG2: DB 0       ; LE NUMERO LOGIQUE DU FICHIER OUVERT

; LA ZONE CONSIGNE POUR LE RETOUR AU NIVEAU DU SYSTEME

5100 0C          ZCSYS: DB 12          ; LONGUEUR DE CETTE CONSIGNE

5101 00          DB  CENTRAL        ; APPEL AU CENTRAL
5102 09          DB  SYSFIC         ; LE GESTIONNAIRE DE FICHIERS
5103 32          DB  SYSTEM         ; RETOUR AU NIVEAU DU SYSTEME

5104 00 00       DB  0,0           ; NOMBRE D'OCTETS
5106 00          DB  0             ; PAGE
5107 00 00       DB  0,0           ; ADRESSE

5109 00          DB  0             ; TYPE D'ORDINATEUR
510A 00          DB  0             ; CODE DE L'APPLICATION

; LE BUFFER DE COMPTE RENDU

510B 0B          CRDUO: DB 8         ; ON DEMANDE UNE RECOPIE DE 8 CARACTERES
510C 00 00 00 00 CRDU: DB 0,0,0,0,0,0,0,0
5110 00 00 00 00

; ICI COMMENCE LE BUFFER POUR LA LECTURE DU FICHIER
; C'EST LA PREMIERE ADRESSE LIBRE

5114          BUFF:

; ICI SE TERMINE LE CODE SPECIFIQUE A NOTRE MINI-APPLICATION

;---- LA FIN DU PROGRAMME

5114          FINPRO:

;---- L'ADRESSE D'EXECUTION

5114 03          DB  3             ; C'EST L'ADRESSE D'EXECUTION
5115 00 00       DW  0             ; LONGUEUR DE L'ENREGISTREMENT
5117 50 00       DW  GOTO         ; L'ADRESSE D'EXECUTION PROPREMENT DITE
5119 00          DB  XPAGE        ; LA PAGE

;---- LA FIN DU FICHIER BINAIRE

511A          FINBIN:

;----

511A          END

```

Exemples de dialogues entre deux postes

Dans ce chapitre, nous donnerons trois exemples d'échanges entre postes qui montrent comment utiliser les sous-programmes qui se trouvent dans le logiciel de communication.

Alors que le chapitre précédent décrivait une petite application complète, nous ne donnerons ici que des extraits significatifs tirés d'une application qui a été réellement écrite et testée mais qui contient un trop grand nombre de choses sans rapport avec notre propos pour être reproduite entièrement.

Les listings de tous ces exemples sont donnés à la fin de ce chapitre : ce sont les fichiers EXEMPLE1.ASM, EXEMPLE2.ASM et EXEMPLE3.ASM. Le fichier COMMUN.ASM contient des éléments (sous-programmes, symboles, variables, etc.) utilisés par les trois exemples.

Tous ces exemples mettent en scène deux postes du réseau que nous appellerons "poste A" et "poste B". Le poste A sera le poste d'où partent les initiatives et le poste B sera donc soumis aux requêtes du poste A.

Exemple 1:

Le poste A lit dans le poste B une zone mémoire de 128 octets. Le numéro du poste B ainsi que l'adresse de la zone mémoire sont fournis par l'utilisateur. La réception des données se fera dans le buffer BUFF situé dans le poste A. Dans cet exemple, le poste B est mis en attente par le poste A.

Exemple 2:

Le poste A remplit avec une valeur fixe une zone mémoire située dans le poste B. Le numéro du poste B, l'adresse de départ, l'adresse d'arrivée et la valeur de remplissage sont fournis par l'utilisateur. A partir de ces données, le poste A construit un segment de code 6809 qu'il donne à exécuter par le poste B. Aucun des deux postes n'est mis en attente dans cet exemple.

Exemple 3:

Cet exemple est plus complexe que les deux précédents. Il montre comment deux postes fonctionnant sous le même langage ou la même application peuvent s'échanger des "programmes" (cf. l'instruction CLONE de BASIC ou la commande PReudre état console de LSE).

Dans cet exemple, le "programme" est le contenu d'une zone mémoire dont l'adresse et la longueur se trouvent dans les variables PRGADR et PRGLGR. Dans le poste A, le "programme" est reçu dans le buffer RECEP. L'adresse de

réception dans le poste A n'est donc pas forcément égale à l'adresse du programme dans le poste B.

Le poste A envoie une demande de programme vers le poste B. Le poste A reste alors en attente de la réponse de B qui se comporte comme un véritable serveur. Le poste B envoie son "programme" au poste A et un compte rendu indiquant que tout s'est bien passé puis sort le poste A de son attente par un ordre "déconnecte-toi".

Lorsque le logiciel de communication du poste B reçoit la demande de programme émanant du poste A, il passe la main à l'application car il ne sait pas traiter le code tâche réseau lui-même. De façon plus précise, il passe la main au segment de code dont l'adresse se trouve dans le pointeur ADTRCTR (\$1FFA - \$1FFB dans la RAM réseau). Ce segment de code commence à l'adresse TRCTR dans notre exemple. Il est exécuté sous interruption et se déroule donc à l'insu du programme principal dans le poste B.

La fonctionnalité "Demande de programme" s'est vu attribuer un code tâche application dépendant du gestionnaire de fichiers bien que ce ne soit pas ce dernier qui réalise l'échange depuis le central. En effet, cette fonctionnalité est importante et pourra se retrouver dans plusieurs langages.

Pour réaliser d'autres catégories d'échanges complexes entre postes, il est nécessaire que chaque application dispose d'un code tâche réseau qui lui soit propre et qui soit différent de celui d'autres applications. A l'intérieur de ce code tâche réseau, chaque catégorie d'échanges pourra être déterminée par un code tâche application. L'attribution des codes tâche réseau est faite par la société Léanord.

; COMMUN.ASM - GILBERT CORNILLIET - 11 01 86

; FICHIER COMMUN AUX EXEMPLES D'ECHANGES ENTRE POSTES

;----- CONSTANTES -----

; LES CODES TACHES RESEAU

00 00	KNOOP	EQU	0	; PAS D'OPERATION
00 01	KOMIAT	EQU	1	; ORDRE DE MISE EN ATTENTE
00 02	KAUTEX	EQU	2	; ORDRE AUTO-EXECUTABLE
00 03	KAFIEC	EQU	3	; ORDRE D'AFFICHER SUR L'ECRAN
00 06	KRCRDU	EQU	6	; ORDRE DE RECOPIER COMPTE RENDU

; POUR BASCULER LES PAGES FORME ET COULEUR DE L'ECRAN

A7 C0	PORT	EQU	0A7C0H	; POUR CHANGER DE PAGES ECRAN
-------	------	-----	--------	-------------------------------

; LES ADRESSES DU LOGICIEL DE COMMUNICATION

A0 25	RESEAU	EQU	\$A025	; APPEL AU RESEAU
A0 2E	EMVE	EQU	\$A02E	; VAS-Y EMETS
A0 31	EMVR	EQU	\$A031	; VAS-Y RECOIS
A0 34	EMDISC	EQU	\$A034	; DECONNECTE-TOI
A0 37	EMAP	EQU	\$A037	; APPEL SOUS ATTENTE

; LA CONSIGNE COURANTE DANS LA RAM RESEAU

1F 5E	CCLGR	EQU	\$1F5E	; LONGUEUR DE LA CONSIGNE COURANTE
1F 5F	CCDST	EQU	\$1F5F	; DESTINATAIRE
1F 60	CCCTR	EQU	\$1F60	; CODE TACHE RESEAU
1F 61	CCCTA	EQU	\$1F61	; CODE TACHE APPLICATION
1F 62	CCNBO	EQU	\$1F62	; NOMBRE D'OCTETS
1F 64	CCPAG	EQU	\$1F64	; PAGE
1F 65	CCADR	EQU	\$1F65	; ADRESSE
1F 68	CCAPPLI	EQU	\$1F68	; LE CODE DE L'APPLICATION
1F 69	CCREST	EQU	\$1F69	; OCTETS DEPENDANT DU CONTEXTE

;----- VARIABLES -----

; LA ZONE CONSIGNE COMMUNE

0000	ZC:			
0000 00	ZCLGR:	DB	0	; LA LONGUEUR DE LA CONSIGNE
0001 00	ZCDST:	DB	0	; LE DESTINATAIRE
0002 00	ZCCTR:	DB	0	; CODE TACHE RESEAU
0003 00	ZCCTA:	DB	0	; CODE TACHE APPLICATION
0004 00 00	ZCNBO:	DW	0	; NOMBRE D'OCTETS
0006 00	ZCPAG:	DB	0	; LA PAGE
0007 00 00	ZCADR:	DW	0	; L'ADRESSE
0009 00	ZCORD:	DB	0	; LE TYPE D'ORDINATEUR
000A 00	ZCAPP:	DB	0	; LE CODE DE L'APPLICATION
000B	ZCREST:	DS	51	; LES OCTETS DEPENDANT DU CONTEXTE ; 51 OCTETS AU MAXIMUM

; BUFFER DE COMPTE RENDU

003E 08 CRDU0: DB B ; ON DEMANDE UNE RECOPIE DE B CARACTERES
003F 00 00 00 00 CRDU: DB 0,0,0,0,0,0,0,0
0043 00 00 00 00

;----- SOUS-PROGRAMMES -----

; <--- PAGECA - PASSAGE EN PAGE CARACTERE

; E) -

; S) SEUL CC MODIFIE
; ON PEUT ACCEDER A LA PAGE CARACTERE DE L'ECRAN

; >---

0047 34 04 PAGECA: PSHS B
0049 F6 A7 C0 LDB PORT
004C CA 01 ORB #1
004E F7 A7 C0 STB PORT
0051 35 B4 PULS B,PC

; <--- COPXYA RECOPIE DE A OCTETS DE X VERS Y

; E) X POINTE SUR LA ZONE MEMOIRE SOURCE
; Y POINTE SUR LA ZONE MEMOIRE DESTINATION
; A CONTIENT LE NOMBRE DE CARACTERES A COPIER

; S) Y POINTE SUR LE PREMIER CARACTERE EN DEHORS DE LA ZONE DESTINATION

; >---

0053 4D COPXYA: TSTA

0054 27 07 CXYA20: BEQ CXYA99 ; PLUS DE CARACTERE A RECOPIER ?
0056 E6 B0 LDB ,X+
0058 E7 A0 STB ,Y+
005A 4A DECA
005B 20 F7 BRA CXYA20

005D 39 CXYA99: RTS

; EXEMPLE1.ASM - GILBERT CORNILLIET - 11 01 86

; LECTURE D'UNE ZONE MEMOIRE DANS UN AUTRE POSTE

00 80

LBUFF EQU 128 ; LA LONGUEUR DE LA ZONE MEMOIRE A LIRE

; <--- LIMEM LECTURE DE LA MEMOIRE DE L'AUTRE POSTE

; E) A CONTIENT LE NUMERO DU POSTE VISE

; X CONTIENT L'ADRESSE DE LA ZONE MEMOIRE DANS L'AUTRE POSTE

; S) -

; NOTE

; SI LE POSTE NE REPOND PAS OU SI L'UTILISATEUR APPUIE SUR <CTRL C>
; PENDANT LES ESSAIS DE COMMUNICATION, LE PROCESSEUR SERA DEROUTE
; SUR LE SOUS-PROGRAMME DONT L'ADRESSE SE TROUVE DANS LE POINTEUR
; ADCTRLC (\$1FFE-\$1FFF DANS LA RAM RESEAU)
; CE SOUS-PROGRAMME N'EST PAS INCLUS DANS CET EXEMPLE.

; >---

0000 B7 00 AD
0003 BF 00 B3

LIMEM: STA ZCDST ; LE DESTINATAIRE
STX ZCADR ; L'ADRESSE DANS L'AUTRE POSTE

0006 C6 0C
0008 F7 00 AC

LDB #12 ; LA LONGUEUR DE LA CONSIGNE
STB ZCLGR

000B 8E 00 B0
000E BF 00 B0

LDX #LBUFF ; LA LONGUEUR DES DONNEES A LIRE
STX ZCNBD

0011 C6 01
0013 F7 00 AE

LDB #KDMJAT ; ON VA METTRE L'AUTRE POSTE EN ATTENTE
STB ZCCTR

0016 8E 00 AC
0019 BD A0 25

LDX #ZC ; ENVOI DE CETTE CONSIGNE VERS L'AUTRE POSTE
JSR RESEAU

001C BD 00 F3

JSR PAGECA ; POUR ACCEDER A LA RAM RESEAU

001F 8E 00 2C
0022 BF 1F 65

LDX #BUFF ; C'EST L'ADRESSE DE RECEPTION CHEZ NOUS
STX CCADR ; DANS LA CONSIGNE COURANTE DE LA RAM RESEAU

0025 BD A0 2E

JSR EMVE ; L'AUTRE POSTE VA NOUS ENVOYER SA MEMOIRE

002B BD A0 34

JSR EMDISC ; ET ON LE LIBERE

; LE BUFFER BUFF CONTIENT MAINTENANT LA ZONE MEMOIRE QUI
; VIENT D'ETRE LUE DANS L'AUTRE POSTE.

002B 39

RTS

; LE BUFFER DU ON LIT LES DONNEES

002C

BUFF DS LBUFF

; <--- FILL REMPLISSAGE D'UNE ZONE MEMOIRE DANS L'AUTRE POSTE

; E) A CONTIENT LE NUMERO DU POSTE VISE
 ; X CONTIENT L'ADRESSE DE DEBUT
 ; Y CONTIENT L'ADRESSE D'ARRIVEE
 ; B CONTIENT LA VALEUR DE REMPLISSAGE

; S) -

; >---

0000	B7 00 3B	FILL:	STA	ZCDST	; LE DESTINATAIRE
0003	BF 00 2B		STX	ADRDEP	; ADRESSE DE DEPART
0006	10 BF 00 32		STY	ADRARR	; ADRESSE D'ARRIVEE
000A	F7 00 2E		STB	VALEUR	; VALEUR DE REMPLISSAGE

; ENVOI D'UNE TRAME AUTO-EXECUTABLE VERS L'AUTRE POSTE
 ; L'AUTRE POSTE EXECUTE LE CODE DE CETTE TRAME IMMEDIATEMENT.
 ; IL N'EST PAS MIS EN ATTENTE ICI.

; ON RECOPIE LE CODE DANS LA ZONE CONSIGNE

000D	BE 00 2A		LDX	#CODE	
0010	10 BE 00 42		LDY	#ZCREST	
0014	B6 0D		LDA	#FCODE-CODE	
0016	BD 00 BA		JSR	COPYXA	
0019	C6 1E		LDB	#30	; LA LONGUEUR DE LA CONSIGNE
001B	F7 00 37		STB	ZCL6R	
001E	C6 02		LDB	#KAUTEX	; CODE TACHE RESEAU : EXECUTION DE CODE
0020	F7 00 39		STB	ZCCTR	
0023	BE 00 37		LDX	#ZC	; ENVOI DE CETTE CONSIGNE VERS L'AUTRE POSTE
0026	BD A0 25		JSR	RESEAU	
0029	39		RTS		

; LE CODE DE LA CONSIGNE AUTO-EXECUTABLE
 ; IL SERA EXECUTE A L'ADRESSE CCREST DANS L'AUTRE POSTE

002A	BE 00 00	CODE:	LDX	##0000	; L'ADRESSE DE DEPART
	00 2B	ADRDEP	EQU	\$-2	
002D	B6 00		LDA	##00	; LA VALEUR
	00 2E	VALEUR	EQU	\$-1	
002F	A7 B0	FLOOP:	STA	,X+	
0031	BC 00 00		CMPX	##0000	; L'ADRESSE D'ARRIVEE
	00 32	ADRARR	EQU	\$-2	
0034	23 F9		BLS	FLOOP	
0036	39		RTS		
0037		FCODE:			

; EXEMPLE3.ASM - GILBERT CORNILLIET - 11 01 86

; ECHANGE DE "PROGRAMME" ENTRE DEUX POSTES

```
-----
00 09      SYSFIC EQU   9           ; CODE TACHE RESEAU DU GESTIONNAIRE DE FICHIERS
00 01      PROG  EQU   1           ; DEMANDE DE PROGRAMME
1F F7      APPLIC EQU  $1FF7       ; L'ADRESSE DU EST RANGE PAR LE CENTRAL LE CODE
                                           ; DE L'APPLICATION LORS DU CHARGEMENT

; <--- CLONE      DEMANDE DU PROGRAMME D'UN AUTRE POSTE

; E)   A CONTIENT LE NUMERO DU POSTE VISE
; S)   EQ <==>  TOUT S'EST BIEN PASSE

; >---

0000 B7 02 10      CLONE: STA      ZCDST           ; DESTINATAIRE
0003 C6 0C          LDB      #12             ; LONGUEUR DE LA CONSIGNE
0005 F7 02 0F      STB      ZCLGR
0008 C6 09          LDB      #SYSFIC        ; LE CODE TACHE RESEAU DU GETIONNAIRE DE
000A F7 02 11      STB      ZCCTR           ; FICHIERS
0000 C6 01          LDB      #PROG          ; DEMANDE DE PROGRAMME
000F F7 02 12      STB      ZCCTA
0012 8E 00 2D      LDX      #RECEP         ; L'ADRESSE DE RECEPTION DU PROGRAMME CHEZ NOUS
0015 8F 02 16      STX      ZCADR
0018 8E 01 90      LDX      #LRECEP        ; LA LONGUEUR MAXIMUM DE CE BUFFER
001B 8F 02 13      STX      ZCNBD

; CETTE CONSIGNE RETOURNE UN COMPTE RENDU. ON LE MET A PRIORI
; A $FF AU CAS OU L'AUTRE POSTE NE REpondRAIT PAS.

; LE COMPTE RENDU, DANS CET EXEMPLE, EST DE LA FORME :

;          CRDU          CODE D'ERREUR (0 <==> PAS D'ERREUR)
;          CRDU+1       LONGUEUR ENVOYEE (SUR DEUX OCTETS)

001E C6 FF          LDB      ##FF
0020 F7 02 4E      STB      CRDU
0023 8E 02 0F      LDX      #ZC              ; ENVOI DE CETTE CONSIGNE A L'AUTRE POSTE
0026 BD A0 25      JSR      RESEAU
0029 F6 02 4E      LDB      CRDU
002C 39             RTS

; RETOUR EQ SI CELA S'EST BIEN PASSE

; ---- LE BUFFER POUR RECEVOIR LE PROGRAMME

01 90      LRECEP EQU   400
002D      RECEP: DS      LRECEP
```

```

; <--- TRCTR - TRAITEMENT DES CODES TACHES RESEAU INCONNUS DE L'EPROM
; E) ON ARRIVE ICI QUAND LE POSTE A RECU UNE CONSIGNE CONTENANT UN
; CODE TACHE RESEAU QU'ELLE NE SAIT PAS EXECUTER. C'EST EN PARTICULIER
; LE CAS LORSQU'UN AUTRE POSTE FONCTIONNANT SOUS L'APPLICATION
; PRESENTE FAIT UNE DEMANDE DE PROGRAMME
; S) ON SERT LA DEMANDE DE PROGRAMME SI ON RECONNAIT LE CODE TACHE RESEAU
; ON NE FAIT RIEN DANS LES AUTRES CAS
; NOTE
; DANS UNE VERITABLE APPLICATION, IL FAUDRAIT TESTER LE CODE TACHE
; RESEAU ET LE CODE TACHE APPLICATION.
; >---

```

```

; ON EST ICI A L'INTERIEUR DU TRAITEMENT D'UNE INTERRUPTION.
; LA SAUVEGARDE DES REGISTRES EST ASSUREE PAR LE LOGICIEL
; DE COMMUNICATION. L'AUTRE POSTE EST EN ATTENTE.
; LA PAGE CARACTERE DE L'ECRAN ET DONC LA RAM RESEAU SONT VISIBLES

```

```

; ON COMMENCE PAR VERIFIER QUE
; LE CODE TACHE RESEAU EST SYSFIC
; LE CODE TACHE APPLICATION EST PROG
; C'EST LA MEME APPLICATION QUI NOUS APPELLE

```

```

01B0 F6 1F 60      TRCTR: LDB   CCCTR           ; LE CODE TACHE RESEAU RECU
01C0 C1 09         CMPB  #SYSFIC
01C2 26 46         BNE   TRCT99

```

```

01C4 F6 1F 61      LDB   CCCTA           ; LE CODE TACHE APPLICATION RECU
01C7 C1 01         CMPB  #PROG
01C9 26 3F         BNE   TRCT99

```

```

01CB F6 1F 6B      LDB   CCAPPLI        ; LE CODE DE L'APPLICATION DE L'AUTRE POSTE
01CE F1 1F F7      CMPB  APPLIC         ; EST-IL EGAL AU NOTRE ?
01D1 26 37         BNE   TRCT99

```

```

; DANS CET EXEMPLE, "ENVOYER LE PROGRAMME" CONSISTE A ENVOYER
; LA ZONE MEMOIRE DONT L'ADRESSE ET LA LONGUEUR SONT CONTENUES
; DANS LES VARIABLES PRGADR ET PRGLGR

```

```

; ON DOIT FAIRE ATTENTION QUE L'AUTRE POSTE NOUS PASSE UNE LONGUEUR
; DE BUFFER MAXIMUM. ON DOIT DONC AJUSTER LA LONGUEUR DES DEUX COTES

```

```

01D3 FC 02 0D      LDD   PRGLGR         ; D = LONGUEUR DE NOTRE PROGRAMME
01D6 10 B3 1F 62   CMPD  CCNBD
01DA 24 0D         BHS   TRCT10        ; SI LA LONGUEUR EST BONNE, PAS D'AJUSTEMENT

```

```

; CAS OU ON ENVOIE A L'AUTRE POSTE MOINS QUE CE QU'IL ATTEND
; ON ENVOIE ALORS UNE CONSIGNE SOUS ATTENTE POUR METTRE
; SES POINTEURS A JOUR

```

```

01DC FD 1F 62      STD   CCNBD          ; LA LONGUEUR SERA LA MEME DANS LES DEUX
; POSTES

```

```

01DF C6 00         LDB   #KNOOP         ; PAS D'AUTRE OPERATION
01E1 F7 1F 60      STB   CCCTR

```

```

01E4 BD A0 37      JSR   EMAP           ; APPEL SOUS ATTENTE
01E7 25 21      BCS   TRCT99

```

; ON ENVOIE MAINTENANT LE PROGRAMME A L'AUTRE POSTE

01E9 BE 02 08 TRCT10: LDX PRGADR ; L'ADRESSE DE NOTRE "PROGRAMME"
01EC BF 1F 65 STX CCADR

01EF BD A0 31 JSR EMVR ; VAS-Y RECOIS
01F2 25 16 BCS TRCT99

; ET, ENFIN, LE COMPTE-RENDU

01F4 C6 86 LDB #KRCRDU.DR.#80 ; RECOPIE APRES LE "DECONNECTE-TOI"
01F6 F7 1F 60 STB CCCTR

01F9 7F 1F 69 CLR CCREST ; PAS D'ERREUR

01FC FC 1F 62 LDD CCNBO ; LA LONGUEUR TRANSMISE
01FF FD 1F 6A STD CCREST+1 ; EST MISE DANS LE COMPTE RENDU
0202 BD A0 37 JSR ENAP ; APPEL SOUS ATTENTE
0205 25 03 BCS TRCT99

0207 BD A0 34 JSR ENDISC ; ON LIBERE L'AUTRE POSTE

020A 39 TRCT99: RTS

020B 80 00 PRGADR: DW \$8000 ; L'ADRESSE ET LA LONGUEUR DE NOTRE
020D 02 50 PRBLGR: DW \$250 ; PROGRAMME (PAR EXEMPLE)